


Advanced X-Sectional Results: Using Paths to Post Process Results

Introduction

This tutorial was created using ANSYS 7.0. The purpose of this tutorial is to create and use 'paths' to provide extra detail during post processing. For example, one may want to determine the effects of stress concentrators along a certain path. Rather than plotting the entire contour plot, a plot of the stress along that path can be made.



In this tutorial, a steel plate measuring 100 mm X 200 mm X 10 mm will be used. Three holes are drilled through the vertical centerline of the plate. The plate is constrained in the y-direction at the bottom and a uniform, distributed load is pulling on the top of the plate.


Preprocessing: Defining the Problem

1. **Give the example a Title**
 - Utility Menu > File > Change Title ...
/title, Use of Paths for Post Processing
2. **Open preprocessor menu**
 - ANSYS Main Menu > Preprocessor
/PREP7
3. **Define Rectangular Ares**
 - Preprocessor > Modeling > Create > Areas > Rectangle > By 2 Corners
BLC4,0,0,200,100
 - Create a rectangle where the bottom left corner has the coordinates 0,0 and the width and height are 200 and 100 respectively.
4. **Create Circles**
 - Preprocessor > Modeling > Create > Areas > Circle > Solid Circle
cyl4,WP X,WP Y,Radius
 - Create three circles with parameters shown below.

Circle	Parameters		
	WP X	WP Y	Radius
1	50	50	10
2	100	50	10
3	150	50	10

5. **Subtract the Circles**

- Preprocessor > Modeling > Operate > Booleans > Subtract > Areas
- First, select the area to remain (ie. the rectangle) and click OK. Then, select the areas to be subtracted (ie. the circles) and click OK.
- The remaining area should look as shown below.

6. **Define the Type of Element**

- Preprocessor > Element Type > Add/Edit/Delete...
- For this problem we will use the PLANE2 (Solid Triangle 6node) element. This element has 2 degrees of freedom (translation along the X and Y axes).
- In the 'Element Types' window, click 'Options...' and set 'Element behavior' to **Plane strs w/thk**

7. **Define Real Constants**

- Preprocessor > Real Constants... > Add...
- In the 'Real Constants for PLANE2' window, enter a thickness of 10.

8. **Define Element Material Properties**

- Preprocessor > Material Props > Material Models > Structural > Linear > Elastic > Isotropic
- In the window that appears, enter the following geometric properties for steel:
 - i. Young's modulus EX: 200000
 - ii. Poisson's Ratio PRXY: 0.3

9. **Define Mesh Size**

- Preprocessor > Meshing > Size Cntrls > ManualSize > Areas > All Areas...
- For this example we will use an element edge length of 5mm.

10. **Mesh the Area**

- Preprocessor > Meshing > Mesh > Areas > Free > click 'Pick All'

Solution Phase: Assigning Loads and Solving

1. **Define Analysis Type**

- Solution > Analysis Type > New Analysis > Static
- ANTYPE, 0


2. **Apply Constraints**

- Solution > Define Loads > Apply > Structural > Displacement > On Lines
- Constrain the bottom of the area in the UY direction.

3. **Apply Loads**

- Solution > Define Loads > Apply > Structural > Pressure > On Lines

- Apply a constant, uniform pressure of -200 on the top of the area.
- The model should now look like the figure below.




4. **Solve the System**
 - Solution > Solve > Current LS
- SOLVE


Postprocessing: Viewing the Results

To see the stress distribution on the plate, you could create a normal contour plot, which would have the distribution over the entire plate. However, if the stress near the holes are of interest, you could create a path through the center of the plate and plot the stress on that path. Both cases will be plotted below on a split screen.

1. **Contour Plot**
 - Utility Menu > PlotCtrls > Window Controls > Window Layout
 - Fill in the 'Window Layout' as seen below



- General Postproc > Plot Results > Contour Plot > Nodal Solu > Stress > von Mises
- The display should now look like this.




To ensure the top plot is not erased when the second plot is created, you must make a couple of changes.

- **Utility Menu > PlotCtrls > Window Controls > Window On or Off.** Turn window 1 'off'.
- To keep window 1 visible during replots, select **Utility Menu > PlotCtrls > Erase Option > Erase Between Plots** and ensure there is no check-mark, meaning this function off.
- To have the next graph plot in the bottom half of the screen, select **Utility Menu > PlotCtrls > Window Controls > Window Layout** and select 'Window 2 > Bottom Half > Do not replot'.

2. **Create Path**


- General PostProc > Path Operations > Define Path > By Location
- In the window, shown below, name the path Cutline and set the 'Number of divisions' to 1000




- Fill the next two window in with the following parameters

Parameters			
Path Point Number	X Loc	Y Loc	Z Loc
1	0	50	0
2	200	50	0


- When the third window pops up, click 'Cancel' because we only enabled two points on the path in the previous step.
3. **Map the Stress onto the Path**
- Now the path is defined, you must choose what to map to the path, or in other words, what results should be available to the path. For this example, equivalent stress is desired.
- General Postproc > Path Operations > Map onto Path
 - Fill the next window in as shown below [Stress > von Mises] and click OK.




- The warning shown below will probably pop up. This is just saying that some of the 1000 points you defined earlier are not on interpolation points (special points on the elements) therefore there is no data to map. This is of little concern though, since there are plenty of points that do lie on interpolation points to produce the necessary plot, so disregard the warning.



4. **Plot the Path Data**
- General Postproc > Path Operations > Plot Path Item > On Geometry
 - Fill the window in as shown below



The display should look like the following. Note, there will be dots on the plot showing node locations. Due to resolution restrictions, these dots are not shown here.



This plot makes it easy to see how the stress is concentrated around the holes.

Command File Mode of Solution

The above example was solved using a mixture of the Graphical User Interface (or GUI) and the command language interface of ANSYS. This problem has also been solved using the ANSYS command language interface that you may want to browse. Copy and paste following code into Notepad or a similar text editor and save it to your computer. Now go to **'File > Read input from...'** and select the file.

finish

```

/clear

/title, Defining Paths
/PREP7

! create geometry

BLC4,0,0,200,100
cyl4,50,50,10
cyl4,100,50,10
cyl4,150,50,10
asba,1,all

et,1,plane2,,,3      ! Plane element
R,1,10               ! thickness of plane
mp,ex,1,200000      ! Young's Modulus
mp,prxy,1,0.3       ! Poisson's ratio

esize,5              ! mesh size
amesh,all            ! area mesh

finish
/solu

! apply constraints
lsel,s,loc,y,0       ! select line for constraint application
dl,all,,UY           ! constrain all DOF's on this face
allsel

! apply loads
allsel               ! restore entire selection
lsel,s,loc,y,100
SFL,all,PRES,-2000/10 ! apply a pressure load on a line
allsel

solve                ! solve resulting system of equations
finish

! plot results
/window,1,top        ! define a window (top half of screen)
/POST1
PLNSOL,S,eqv,2,1     ! plot stress in xx direction (deformed and undeformed edge)
/window,1,off
/noerase

/window,2,bot        ! define a window (bottom half of screen)
nset,all             ! define nodes to define path
nset,s,loc,y,50      ! choose nodes half way through structure
path,cutline,2,,1000 ! define a path labeled cutline
ppath,1,,0,50        ! define endpoint nodes on path
ppath,2,,200,50
PDEF,,S,eqv,AVG      ! calculate equivalent stress on path
nset,all
PLPAGM,SEQV,200,NODE ! show graph on plot with nodes

```

Reference

<http://www.mece.ualberta.ca/tutorials/ansys>

v